
Records – a user defined type 
In our periodic table project, we want to store the following information in arrays 

 Element Name (string) 
 Element Symbol (string) 
 Element Atomic Mass (single) 

 
Now, we could declare three arrays each with indices from 1 .. 118.   
 
We will introduce the concept of a record instead with this example involving solar system data.  We need to insert a module into our project and put our record 
definition in there: 
'User defined type to store the data for a single planet 
Public Type Planet 
Planet_Name As String 
Orbital_Radius As Single 
Orbital_period As Single 
End Type 
'array of planets 
Public Planets(1 To 9) As Planet 
 
We will be able to access the array of records Planets() in our main program under the UserForm. 
 
Consider the data from the textfile OrbitalData.txt 
Notice that the data in OrbitalData.txt is a comma separated values (CSV) format.  We don’t need to worry about the first four lines of this data, so the first four 
lines don’t have commas.  The commas define the columns of data in this textfile.  There are 10 columns for each planet 
 
                        Distance O_Period 
Name       #     Orbits (000 km)   (days)    Incl Eccen  Discoverer   Date   A.K.A. 
---------  ---- ------- --------  --------  ----- -----  ----------   ----  ------- 
Sun        -       -       -          -        -     -     -           -    Sol(0) 
Mercury,   I,   Sun,       57910,   87.97,   7.00, 0.21,    -,          -,  (0) 
Venus,     II,  Sun,      108200,  224.70,   3.39, 0.01,    -,          -,  (0) 
 
 
  



Option Explicit 
Private Sub LoadData() 
 
'we only want to load the planet name, planet orbital radius and planet orbital period 
Dim strTmp1 As String, strTmp2 As String, strTmp3 As String, strTmp4 As String, strTmp5 As String 
Dim strTmp6 As String, strTmp7 As String 
Dim strPath As String 
Dim i As Integer 
Dim intFile As Integer 'to store the file handle (a number) 
strPath = ThisDocument.Path + "\OrbitalData.txt"  'set location to input file 
 
intFile = FreeFile 'automatically assign a free file number to intFile 
 
Open strPath For Input As intFile  'we associate the actual file (strPath) with the file number (intFile) 
 
'we need to skip the first four lines of data 
For i = 1 To 4 
   Line Input #intFile, strTmp1 'read in a whole line from the textfile into strTmp1 
   ListBox1.AddItem strTmp1 
Next 
i = 0 
Do Until EOF(intFile) 
   i = i + 1 
   Input #intFile, Planets(i).Planet_Name, strTmp1, strTmp2, Planets(i).Orbital_Radius, _ 
             Planets(i).Orbital_period, strTmp3, strTmp4, strTmp5, strTmp6, strTmp7    
   Planets(i).Orbital_Radius = Planets(i).Orbital_Radius * 1000 'need to multiply by 1000 
   ListBox1.AddItem Planets(i).Planet_Name + vbTab + Str(Planets(i).Orbital_Radius) + vbTab + _ 
                    Str(Planets(i).Orbital_period) 
Loop 
Close (intFile) 'should always close file when done! 
End Sub 
 
Private Sub UserForm_Initialize() 
Call LoadData 
End Sub 


